

SYNTAX AND SEMANTICS OF BABEL-17

STEVEN OBUA

CONTENTS

1. Introduction	1
2. Reference Implementation	2
3. Lexical Matters	2
4. Overview of Built-in Types	3
5. Objects and Functions	3
6. Exceptions	4
7. Laziness and Concurrency	5
8. Lists and Vectors	6
9. CExprs	6
10. Pattern Matching	7
11. Non-Determinism in Babel-17	8
12. Block Expressions	9
13. Anonymous Functions	12
14. Object Expressions	13
15. Boolean Operators	14
16. Order	15
17. Sets and Maps	16
18. Reals and Interval Arithmetic	17
19. Syntactic Sugar	18
20. Memoization	18
21. Linear Scope	19
22. Record Updates	21
23. With	22
24. Loops	22
25. Modules and Types	24
26. Pragmas	24
27. Unit Tests	24
28. Standard Library (To Be Updated for v0.3)	24
28.1. Collections and Collectors	24
28.2. Type Conversions	24
28.3. Integer, Infinity	25
28.4. String	25
28.5. List and Vector	26
28.6. Set	26

Date: May 21, 2011.

28.7. Map	26
28.8. Object	26
29. What's Next?	26

1. INTRODUCTION

The first question that someone who creates a new programming language will hear from others inevitably is: Why another programming language? Are there not already enough programming languages out there that you can pick from?

I always wanted a programming language

- in which I can express myself compactly and without jumping through too many hoops,
- that is easy to understand and to learn,
- in which purely functional programming is the default and not the exception,
- that supports exceptions,
- that has pattern matching,
- that has built-in support for laziness and concurrency,
- that has built-in support for purely functional lists, vectors, sets and maps,
- that seamlessly marries structured programming constructs like loops with purely functional programming,
- that supports objects, modules and data encapsulation,
- that has mature implementations for all major modern computing platforms and can (and should!) be used for real-world programming,
- that has a simple mechanized formal semantics,
- that is beautiful.

There is no such language out there. Therefore I decided to create one. When Babel-17 will have reached v1.0 then the above goals will (mostly) have been achieved. The version described in this document is Babel-17 v0.3, so there is still some distance to go.

Babel-17 is not a radically new or revolutionary programming language. It picks those raisins I like out of languages like Standard ML, Scala, Alice ML, Java, Javascript, Erlang, Haskell, Lisp, Clojure and Mathematica, and tries to evolve them into a beautiful new design.

Babel-17 is a functional language. It is not a pure functional language because it contains sources of non-determinism like the ability to pick one of two values randomly. If you take these sources of non-determinism out of the language (there are currently only two such sources, **choose** and **random**), it becomes pure.

In Babel-17, every value has a type. These types are not checked statically, but only dynamically.

Babel-17 is also object-oriented in that many aspects of Babel-17 can be explained in terms of sending messages to objects.

The default evaluation order of Babel-17 is strict. But optional laziness is built into the heart of Babel-17, also, and so is concurrency.

Babel-17 has pattern matching and exception handling. Both are carefully integrated with each other and with the laziness and concurrency capabilities of Babel-17.

Furthermore, Babel-17 establishes the paradigm of *purely functional structured programming* via the notion of linear scope.

In this document I specify Babel-17 in a mostly informal style: formal notation is sometimes used when it supports clarity.

2. REFERENCE IMPLEMENTATION

At <http://babel-17.com> you will find a reference implementation of Babel-17 v0.3 that fully supports the language as described in this document. There you will also find a Netbeans plugin that has syntax and error-highlighting for Babel-17 programs. As you find your way through this document, it might be helpful to directly try out Babel-17 in Netbeans to see if the interpreter behaves as you would guess from this spec.

3. LEXICAL MATTERS

Babel-17 source code is always written UTF-8 encoded. If it ain't UTF-8, it ain't Babel-17.

Constructors are alphanumeric strings which can also contain underscores, and which must start with a capital letter. Identifiers are alphanumeric strings which can also contain underscores, and which must start with a non-capital letter. The distinction between two constructors or two identifiers is not sensitive to capitalization.

These are the keywords of Babel-17:

begin	end	object	with	if	then	else	elseif
while	for	do	choose	random	yield	match	case
as	val	def	in	exception	lazy	concurrent	memoize
to	downto	true	false	nil	infinity	force	this
try	catch	typedef	typeof	module	private	import	unittest

Note that keywords are written always in lower case. For example, `bEGIN` is not a legal keyword, but it also isn't an identifier (because of case insensitivity). Also note that `Begin` and `BEGIN` denote the same legal constructor. When we talk about identifiers in the rest of the paper, we always mean identifiers which are not keywords.

You can write down decimal, hexadecimal, binary and octal integers in Babel-17. The following notations all denote the same integer:

15 0xF 0b1111 0o17

There is also a decimal syntax for floating point numbers in Babel-17:

15.0 1.5E1 1.5E+1 150E-1 15e0

Strings start and end with a quotation mark. Between the quotation mark, any valid UTF-8 string is allowed that does not contain newline, backslash or quotation mark characters, for example:

`"hello world" "$1, 2%" "1 <= 4! <= 5^2"`

Newline, backslash and quotation mark characters can be written down in escaped form. Actually, any unicode character can be written down in its escaped form via the hexadecimal value of its codepoint:

TABLE 1. Further Unicode Symbols

Unicode Hex Code	Display	ASCII Equivalent
2261	≡	==
2262	≠	!=
2264	≤	<=
2265	≥	>=
2227	∧	&
2228	∨	
00AC	¬	!
2237	::	::
2192	→	->
21D2	⇒	=>
2026

String consisting only of a quotation mark	"\""
String consisting only of a backslash	"\\"
String consisting only of a line feed	"\n"
String consisting only of a carriage return	"\r"
String consisting only of line feed via 16-bit escaping	"\u000A"
String consisting only of line feed via 32-bit escaping	"\U0000000A"

Multi-line comments in Babel-17 are written between matching pairs of `#(` and `)#` and can span several lines. Single-line comments start with `##`.

Pragmas start with `#`, just like comments. There are the following pragmas: `#assert`, `#log`, `#print`, `#profile`.

Finally, here are all ASCII special symbols that can occur in a Babel-17 program:

```
=  ==  !=  <  <=  >  >=  +  -  *  /  ^  ;
|  &  !  ++  --  **  //  ,  ::  ->  =>  ?  ...
(  )  [  ]  {  }  .  :  ~  :>
```

Table 1 lists all other Unicode symbols that have a meaning in Babel-17.

4. OVERVIEW OF BUILT-IN TYPES

Each value in Babel-17 has a unique type. All built-in types are depicted in Table 2. Often we use the name of a type to refer to this type in this specification, but the representation of it in actual Babel-17 is given in the type column. The type of a value v can be obtained via

`typeof v`

5. OBJECTS AND FUNCTIONS

All values in Babel-17 are objects. This means that you can send messages to them. Many values in Babel-17 are functions. This means that you can apply them to an argument, yielding another value. Note that for a value to be an object, it does not need to have type `obj`, and for a value to be a function, it does not need to have type `fun`.

The result of sending message m to object v is written as

TABLE 2. Built-in Types of Babel-17

Name	Type	Description
<i>Integer</i>	int	the type of arbitrary size integer numbers
<i>Real</i>	real	the type of real numbers
<i>Boolean</i>	bool	the type consisting of the values true and false
<i>String</i>	string	the type of valid Unicode strings
<i>List</i>	list	the type of lists
<i>Vector</i>	vect	the type of vectors / tuples
<i>Set</i>	set	the type of sets
<i>Map</i>	map	the type of maps
<i>CExpr</i>	cexpr	the type of constructed expressions
<i>Object</i>	obj	the type of user-defined objects
<i>Function</i>	fun	the type of functions
<i>PersistentException</i>	exc	the type of persistent exceptions
<i>DynamicException</i>		the type of dynamic exceptions
<i>Type</i>	type	the type of types

$$v . m$$

Here m is an identifier.

The result of applying function f to value x is written as

$$f x$$

Note that $f x$ is equivalent to

$$(f.\text{apply_}) x$$

Therefore any object that responds to the `apply_` message can act as a function. In the above we could also leave out the brackets because sending messages binds stronger than function application.

Repeated function application associates to the left, therefore

$$f x y$$

is equivalent to $(f x) y$.

6. EXCEPTIONS

Exception handling in Babel-17 mimicks exception handling mechanisms like those which can be found in Java or Standard ML, while adhering at the same time to the functional paradigm.

There are two types of exceptions: *PersistentException* and *DynamicException*. The difference is that a *PersistentException* can be treated as part of any data structure, and is passed around just like any other value, while a *DynamicException* can never be part of a data structure and has special evaluation semantics.

Exceptions in Babel-17 are not that special at all but mostly just another type of value. Let us write *PersistentException* v for a *PersistentException* with parameter v , and *DynamicException* v for a *DynamicException* with parameter v . We also write *Exception* v for an exception with parameter v . The parameter v is a *non-exceptional*

value; this means that v can be any Babel-17 value except one of type *DynamicException*. Note that a value of type *PersistentException* is therefore non-exceptional.

The following deterministic rules govern how exceptions behave with respect to sending of messages, function application, laziness and concurrency:

$$\begin{aligned}
(\text{Exception } v).m &\rightsquigarrow \text{DynamicException } v \\
(\text{DynamicException } v) x &\rightsquigarrow \text{DynamicException } v \\
f (\text{DynamicException } v) &\rightsquigarrow \text{DynamicException } v \text{ where } f \text{ is non-exceptional} \\
(\text{PersistentException } v) g &\rightsquigarrow \text{DynamicException } v \text{ where } g \text{ is non-exceptional} \\
\text{exception } v &\rightsquigarrow \text{DynamicException } v \\
\text{lazy } (\text{Exception } v) &\rightsquigarrow \text{PersistentException } v \\
\text{concurrent } (\text{Exception } v) &\rightsquigarrow \text{PersistentException } v \\
\text{force } v &\rightsquigarrow v \text{ for all } v, \text{ including exceptions}
\end{aligned}$$

Exceptions in Babel-17 are created with the expression **exception** v . Creating exceptions in Babel-17 corresponds to *raising* or *throwing* exceptions in other languages. Catching an exception can be done via **match** or via **try-catch**. Both constructs will be described later.

In the next section, we will describe the **lazy** and **concurrent** expressions of Babel-17. They are the reason why exceptions are divided into dynamic and persistent ones.

7. LAZINESS AND CONCURRENCY

The default evaluation mechanism of Babel-17 is strict. This basically means that the arguments of a function are evaluated before the function is applied to them. Babel-17 has two constructs to change this default behaviour.

The expression

lazy e

is not evaluated until it is actually needed. When it is needed, it evaluates to whatever e evaluates to, with the exception of dynamic exceptions, which are converted to persistent ones.

The expression

concurrent e

also evaluates to whatever e evaluates to, again with the exception of dynamic exceptions which are converted to persistent ones. This evaluation will happen concurrently.

One could think that apart from obvious different performance characteristics, the expressions **lazy** e and **concurrent** e are equivalent. This is not so. If e is a non-terminating expression then, even if the value of **concurrent** e is never needed during program execution, it might still lead to a non-terminating program execution. In other words, the behaviour of **concurrent** e is unspecified for non-terminating e .

Sometimes you do not want to proceed until an expression has been completely evaluated so that it does not contain any lazy or concurrent subcomponents any more. In those situations you use the expression

force e

which evaluates to e . So semantically, **force** is just the identity function.

We mentioned before that lazy and concurrent expressions are the reason why exceptions are divided into dynamic and persistent ones. To motivate this, look at the expression

`fst (0, lazy (1 div 0))`

Here the function `fst` is supposed to be a function that takes a pair and returns the first element of this pair. So what would above expression evaluate to? Obviously, `fst` does not need to know the value of the second element of the pair as it depends only on the first element, so above expression evaluates just to 0. Now, if **lazy** was semantically just the identity function, then we would have

$$\begin{aligned} 0 &= \text{fst } (0, \text{lazy } (1 \text{ div } 0)) = \text{fst } (0, 1 \text{ div } 0) = \text{fst } (0, \text{exception } \text{DomainError}) \\ &= \text{fst } (\text{exception } \text{DomainError}) = \text{exception } \text{DomainError} \end{aligned}$$

Obviously, 0 should not be the same as an exception, and therefore **lazy** cannot be the identity function, but converts dynamic exceptions into persistent ones. For a dynamic exception e the equation

$$(0, e) = e$$

holds. For a persistent exception e this equation does not hold, and therefore the above chain of equalities is broken.

8. LISTS AND VECTORS

For $n \geq 0$, the expression

$$[e_1, \dots, e_n]$$

denotes a *list* of n elements.

The expression

$$(e_1, \dots, e_n)$$

denotes a *vector* of n elements, at least for $n \neq 1$. For $n = 1$, there is a problem with notation, though, because (e) is equivalent to e . Therefore there is the special notation $(e,)$ for vectors which consist of only one element.

The difference between lists and vectors is that they have different performance characteristics for the possible operations on them. Lists behave like simply linked lists, and vectors behave like arrays. Note that all data structures in Babel-17 are immutable.

Another way of writing down lists is via the right-associative `::` constructor:

$$h::t$$

Here h denotes the *head* of the list and t its *tail*. Actually, note that the expression $[e_1, \dots, e_n]$ is just syntactic sugar for the expression

$$e_1::e_2::\dots::e_n::[]$$

Dynamic exceptions cannot be part of a list but are propagated:

$$\begin{aligned} (\text{DynamicException } v)::t &\rightsquigarrow \text{DynamicException } v \\ h::(\text{DynamicException } v) &\rightsquigarrow \text{DynamicException } v \quad \text{where } h \text{ is non-exceptional} \end{aligned}$$

Note that when the tail t is not a list, we identify $h::t$ with $h::t::[]$.

9. CEXPRs

A *CExpr* is a constructor c together with a parameter p , written $c\ p$. It is allowed to leave out the parameter p , which then defaults to **nil**. For example, **HELLO** is equivalent to **HELLO nil**. A constructor c cannot have a dynamic exception as its parameter, therefore we have:

$$c\ (\textit{DynamicException}\ v) \rightsquigarrow \textit{DynamicException}\ v$$

10. PATTERN MATCHING

Maybe the most powerful tool in Babel-17 is pattern matching. You use it in several places, most prominently in the **match** expression which has the following syntax:

```

match  $e$ 
  case  $p_1 \Rightarrow b_1$ 
   $\vdots$ 
  case  $p_n \Rightarrow b_n$ 
end

```

Given a value e , Babel-17 tries to match it to the patterns p_1, p_2, \dots and so on sequentially in that order. If p_i is the first pattern to match, then the result of **match** is given by the block expression b_i . If none of the pattern matches then there are two possible outcomes:

- (1) If e is a dynamic exception, then the value of the match is just e .
- (2) Otherwise the result is a dynamic exception with parameter NoMatch.

A few of the pattern constructs incorporate arbitrary value expressions. When these expressions raise exceptions, the whole pattern they are contained in is rendered non-matching.

So what does a pattern look like? Table 3 and Table 4 list all ways of building a pattern.

In this table of pattern constructions we use the δ -pattern δ . This pattern stands for "the rest of the entity under consideration" and can be constructed by the following rules:

- (1) The ellipsis \dots is a δ -pattern that matches any rest.
- (2) If δ is a δ -pattern, and x an identifier, then $(x\ \text{as}\ \delta)$ is a δ -pattern.
- (3) If δ is a δ -pattern, and e an expression, then $(\delta\ \text{if}\ e)$ is a δ -pattern.

Note that pattern matching does not distinguish between vectors and lists. A pattern that looks like a vector can match a list, and vice versa.

Besides the **match** construct, there is also the **try** construct. While **match** can handle exceptions, most of the time it is more convenient to use **try** for this purpose. The syntax is

```

try
   $s_1$ 
   $\dots$ 
   $s_m$ 
catch
  case  $p_1 \Rightarrow b_1$ 

```


TABLE 3. General Patterns

Syntax	Description
-	the underscore symbol matches anything but a dynamic exception
x	an identifier x matches anything but a dynamic exception and binds the matched expression to x
$(x \text{ as } p)$	matches p , and binds the successfully matched value to x ; the match fails if p does not match or if the matched value is a dynamic exception
z	a number z , like 0 or 42 or -10 or infinity or -infinity , matches just that number z
$c \ p$	matches a <i>CExpr</i> with constructor c if the parameter of the <i>CExpr</i> matches p ; instead of $c \ -$ you can just write c
s	a string s , like "hello", matches just that string s
(p)	same as p
$(p \text{ if } e)$	matches any non-exceptional value that matches p , but only if e evaluates to true ; identifiers bound in p can be used in e
$(\text{val } e)$	matches any non-exceptional value which is equivalent to e
$(f \ ? \ p)$	f is applied to the value to be matched; the match succeeds if the result of the application matches p
$(e \ ?)$	short for $(e \ ? \ \text{true})$
$\{m_1 = p_1, \dots, m_n = p_n\}$	matches a value of type obj that has exactly the messages m_1, \dots, m_n such that the message values match the given patterns
$\{m_1 = p_1, \dots, m_n = p_n, \delta\}$	matches a value of type obj that has the messages m_1, \dots, m_n such that the message values match the given patterns
nil	matches the empty object
exception p	matches any exception such that its parameter matches p
$(p : t)$	matches anything that has type t and matches p
$(p : (e))$	matches anything that matches p and has the type that e evaluates to

```

⋮
case  $p_n \Rightarrow b_n$ 
end

```

The meaning of the above is identical to the meaning of

```

match
begin
   $s_1$ 
  ...

```

TABLE 4. Collection Patterns

Syntax	Description
$[p_1, \dots, p_n]$	matches a list/vector consisting of $n \geq 0$ elements, such that element e_i of the list is matched by pattern p_i
(p_1, \dots, p_n)	matches a vector/list consisting of $n = 0$ or $n \geq 2$ elements, such that element e_i of the list is matched by pattern p_i
$[p_1, \dots, p_n, \delta]$	matches a list/vector consisting of at least $n \geq 1$ elements, such that the first n elements e_i of the list/vector are matched by the patterns p_i
$(p_1, \dots, p_n, \delta)$	matches a vector/list consisting of at least $n \geq 1$ elements, such that the first n elements e_i of the vector/list are matched by the patterns p_i
$(p,)$	matches a vector/list consisting of a single element that matches the pattern p .
$(h::t)$	matches a non-empty list/vector such that h matches the head of the list/vector and t its tail
$\{p_1, \dots, p_n\}$	see section 17
$\{p_1, \dots, p_n, \delta\}$	see section 17
$\{q_1 \rightarrow p_1, \dots, q_n \rightarrow p_n\}$	see section 17
$\{q_1 \rightarrow p_1, \dots, q_n \rightarrow p_n, \delta\}$	see section 17
$\{->\}$	see section 17
(for p_1, \dots, p_n end)	see section 24
(for p_1, \dots, p_n, δ end)	see section 24

```

      sm
    end
  case (exception p1) => b1
    :
  case (exception pn) => bn
  case x => x
end

```

except for the fact that the latter expression might not be legal Babel-17 because of linear scoping violations.

11. NON-DETERMINISM IN BABEL-17

One source of non-determinism in Babel-17 is *probabilistic* non-determinism. The expression

```
random n
```

returns for an integer $n > 0$ a number between 0 and $n - 1$ such that each number between 0 and $n - 1$ has equal chance of being returned. If n is a dynamic exception, then this exception is propagated, if n is a non-exceptional value that is not an integer > 0 then the result is an exception with parameter `DomainError`.

The other source of non-determinism is choice. The expression

choose l

takes a non-empty collection l and returns a member of the collection. The choice operator makes it possible to optimize evaluation in certain cases. For example, in

choose (**concurrent** a , **concurrent** b)

the evaluator could choose the member that evaluates quicker.

Babel-17 has been designed such that if you take the above two constructs out of the language, it becomes purely functional. If in your semantics of Babel-17 you replace the concept of value by the concept of a probability distribution over values and a set of values, you might be able to view Babel-17 as a purely functional language even *including* **random** and **choose**.

12. BLOCK EXPRESSIONS

So far we have only looked at expressions. We briefly mentioned the term *block expressions* in the description of the **match** function, though. We will now introduce and explain block expressions.

Block expressions can be used in several places as defined by the Babel-17 grammar. For example, they can be used in a **match** expression to define the value that corresponds to a certain case. But block expressions can really be used just everywhere where a normal expression is allowed:

begin
 b
end

is a normal expression where b is a block expression. A block expression has the form

s_1
 \vdots
 s_n

where the s_i are *statements*. In a block expression both newlines and semicolons can be used to separate the statements from each other.

Statements are Babel-17's primary tool for introducing identifiers. There are several kinds of statements. Three of them will be introduced in this section.

First, there is the **val**-statement which has the following syntax:

val $p = e$

Here p is a pattern and e is an expression. Its meaning is that first e gets evaluated. If this results in a dynamic exception, then the result of the block expression that the **val**-statement is part of will be that dynamic exception. Otherwise, the result of evaluating e is matched to p . If the match is successful then all identifiers bound by the match can be used in later statements of the block expression. If the match fails, then the value of the containing block expression is the dynamic exception NoMatch.

Second, there is the **def**-statement which obeys the following syntax for defining the identifier id :

def $id\ arg = e$

TABLE 5. Legal and illegal definitions

val x = y	def x = y	val x = y	def x = y
val y = 0	val y = 0	def y = 0	def y = 0
<i>illegal</i>	<i>illegal</i>	<i>legal</i>	<i>legal</i>

The *arg* part is optional. If *arg* is present, then it must be a pattern. Let us call those definitions where *arg* is present a *function definition*, and those definitions where *arg* is not present a *simple definition*.

Per block expression and identifier *id* there can be either a single simple definition, or there can be several function definitions. If there are multiple function definitions for the same identifier in one block expression, then they are bundled in that order to form a single function.

The defining expressions in **def**-statements can recursively refer to the other identifiers defined by **def**-statements in the same block expression. This is the main difference between definitions via **def** and definitions via **val**. Only those **val**-identifiers are in scope that have been defined *previously*, but **def**-identifiers are in scope throughout the whole block expression. Table 5 exemplifies this rule.

Let us assume that a block expression contains multiple function definitions for the same identifier *f*:

```

def f p1 = e1
  ⋮
def f pn = en

```

Then this is (almost) equivalent to

```

def f x =
  match x
  case p1 => e1
    ⋮
  case pn => en
end

```

where *x* is fresh for the *p_i* and *e_i*. The slight difference between the two notations is that arguments that match none of the patterns will result in a `DomainError` exception for the first notation, and in a `NoMatch` exception for the second.

While definitions only define a single entity per block and identifier, it is OK to have multiple **val**-statements for the same identifier in one block expression, for example like that:

```

val x = 1
val x = (x, x)
x

```

The above block expression evaluates to (1,1); later **val**-definitions overshadow earlier ones. But note that neither

```

val x = 1

```

```
def x = 1
```

nor

```
def x = 1
```

```
val x = 1
```

are legal.

Another difference between **val** and **def** is observed by the effects of non-determinism:

```
val x = random 2
```

```
(x, x)
```

will evaluate either to (0, 0) or to (1, 1). But

```
def x = random 2
```

```
(x, x)
```

can additionally also evaluate to (0, 1) or to (1, 0) because x is evaluated each time it is used.

Let us conclude this section by describing the third kind of statement. It has the form

```
yield e
```

where e is an expression. There is also an abbreviated form of the **yield**-statement which we have already used several times in this section:

```
 $e$ 
```

The semantics of the **yield**-statement is that it appends a further value to the value of the block expression it is contained in. Block expressions have a value just like all other expressions in Babel-17. It is obtained by "concatenating" all values of the **yield**-statements in a block expression. The value of

```
begin
```

```
end
```

is the empty vector (). The value of

```
begin
```

```
yield a
```

```
end
```

is a . The value of

```
begin
```

```
yield a
```

```
yield b
```

```
end
```

is the vector (a, b) of length 2 and so on.

In this section we have introduced the notions of block expressions and statements. Their full power will be revealed in later sections of this document.

13. ANONYMOUS FUNCTIONS

So far we have seen how to define named functions in Babel-17. Sometimes we do not need a name for a certain function, for example when the code that implements this function is actually just as easy to understand as any name for the function. We already have the tools for writing such nameless, or anonymous, functions:

```
begin
  def sqr x = x * x
  sqr
end
```

is an expression denoting the function that squares its argument. There is a shorter and equivalent way of writing down the above:

$$x \Rightarrow x * x$$

In general, the syntax is

$$p \Rightarrow e$$

where p is a pattern and e an expression. The above is equivalent to

```
begin
  def f p = e
  f
end
```

where f is fresh for p and e .

There is also a syntax for anonymous functions which allows for several cases:

```
(case p1 => b1
  ⋮
  case pn => bn)
```

is equivalent to

```
begin
  def f p1 = begin b1 end
  ⋮
  def f pn = begin bn end
  f
end
```

where f is fresh for the p_i and b_i .

14. OBJECT EXPRESSIONS

Object expressions have the following syntax:

```
object
  s1
  ⋮
  sn
end
```

The s_i are statements. Those s_i that are **def** statements define the set of messages that the object responds to.

Often you do not want to create objects from scratch but by modifying other already existing objects:

```

object + parents
   $s_1$ 
   $\vdots$ 
   $s_n$ 
end

```

The expression `parents` must evaluate to either a list, a vector or a set. The members of `parents` are considered to be the parents of the newly created object, in the order induced by the collection. The idea is that the created object not only understands the messages defined by the s_i , but also the messages of the parents. Messages defined via s_i shadow messages of the parents. The messages of an earlier parent shadow the messages of a later one.

The keyword **this** can be used only in **def**-statements of an object and points to the object that the original message has been sent too. There is also a way to denote record-like objects:

$$\{ m_1 = v_1, \dots, m_n = v_n \}$$

This is equivalent to:

```

begin
  val ( $w_1, \dots, w_n$ ) = ( $v_1, \dots, v_n$ )
  object
    def  $m_1 = w_1$ 
    ...
    def  $m_n = w_n$ 
  end
end

```

The empty object is denoted by

```
nil
```

which is equivalent to

```
object end
```

15. BOOLEAN OPERATORS

Babel-17 provides the usual boolean operators. They are just syntactic sugar for certain **match** expressions; the exact translations are given in Table 6. Babel-17 also has **if**-expressions with the following syntax:

```

if  $b_1$  then
   $e_1$ 
elseif  $b_2$  then
   $e_2$ 
   $\vdots$ 

```

TABLE 6. Boolean Operators

! a	a & b	a b
<pre> match a case true => false case false => true case _ => exception DomainError end </pre>	<pre> match a case true => match b case true => true case false => false case _ => exception DomainError end case false => false case _ => exception DomainError end </pre>	<pre> match a case false => match b case true => true case false => false case _ => exception DomainError end case true => true case _ => exception DomainError end </pre>

```

elseif  $b_n$  then
   $e_n$ 
else
   $e_{n+1}$ 
end

```

The **elseif**-branches are optional. They can be eliminated in the obvious manner via nesting, so that we only need to give the semantics for the expression

```

if b then  $e_1$  else  $e_2$  end

```

The meaning of above expression is defined to be

```

match b
  case true =>  $e_1$ 
  case false =>  $e_2$ 
  case _ => exception DomainError
end

```

Actually, the **else** branch is also optional. The notation

```

if b then  $e$  end

```

is shorthand for

```

if b then  $e$  else end

```

16. ORDER

Babel-17 has a built-in partial order which is defined in terms of the operator \sim . The expression $a \sim b$ returns an *Integer*; the usual relational operators are defined in Table 7. Note that a and b are unrelated iff $a \sim b$ is less than -1 or greater than 1 .

It is possible to chain relational operators like this:

TABLE 7. Relational Operators

Syntax	Semantics
$a < b$	$(a \sim b) < 0$
$a == b$	$(a \sim b) == 0$
$a > b$	$(a \sim b) > 0$
$a \leq b$	$(a \sim b) \leq 0$
$a \geq b$	$(a \sim b) \geq 0$
$a != b$	$(a \sim b) != 0$

$$a \leq b \leq c > d != e$$

Intuitively, the above means

$$a \leq b \ \& \ b \leq c \ \& \ c > d \ \& \ d != e .$$

Note that we always evaluate the operands of relational operators, even chained ones, only once. For example, the precise semantics of $a \leq b \leq c \leq d \leq e$ is

```

begin
  val t = a
  val u = b
  t <= u &
  begin
    val v = c
    u <= v &
    begin
      val w = d
      v <= w & w <= e
    end
  end
end
end

```

In the above, t , u , v and w are supposed to be fresh identifiers. Also note that if there are operands that are dynamic exceptions, then the result of a comparison is a dynamic exception with the same parameter as the first such operand (from left to right).

It is possible that two values a and b are not related with respect to the built-in order. In this case, $a \sim b$ throws an *Unrelated*-exception. Taking this into account, the definition of $a \text{ op } b$ for $op \in \{<, >, \leq, \geq, ==\}$ is

```

match a ~ b
case (u if u op 0) => true
case _ : int => false
case (exception Unrelated) => false
end

```

The built-in partial order has the following properties:

- values of type **type** are totally ordered by their name
- $a \sim b$ is equivalent to

$$\text{match } (\text{typeof } a) \sim (\text{typeof } b)$$

```

case 0 =>
  match a.compare_ b
    case u : int => u
    case e => exception CompareError e
  end
case _ => exception Unrelated
end

```

- persistent exceptions are partially ordered by their parameter
- booleans are totally ordered and **false** < **true** holds
- integers are totally ordered in the obvious way
- strings are totally ordered via lexicographic ordering
- lists are partially ordered by the lexicographic ordering
- vectors are partially ordered by the lexicographic ordering
- constructed expressions are partially ordered by representing them by the pair consisting of constructor and parameter
- $f \neq g$ for all values f and g of type **fun**
- sets are partially ordered by first comparing their sizes, and then their elements
- maps are partially ordered by first comparing their sizes, then their keys, and then their corresponding values

17. SETS AND MAPS

Sets and maps are built into Babel-17. For example, the set consisting of 3, 42 and 15 can be written as

$$\{42, 15, 3\}$$

Sets are always sorted. The sorting order is Babel-17's built-in partial order, and every set forms a totally ordered subdomain of this partial order. Future versions of Babel-17 might allow the set order to be different from the built-in partial order. Adding or removing an element e from a set S is only well-defined when the elements of S together with e are totally ordered by the partial order of S . The same holds for testing if an element is in a set.

Maps map finitely many keys to values. For example, the map that maps 1 to 2 and 4 to 0 is written as

$$\{1 \rightarrow 2, 4 \rightarrow 0\}$$

The empty map is denoted by

$$\{\rightarrow\}$$

Maps also have always a partial order associated with them, in the current version of Babel-17 this is always Babel-17's built-in partial order. Operations on maps are only well-defined if all keys of the map together with all other involved keys are totally ordered by the associated order.

Pattern matching is available also for sets and maps. The pattern

$$\{p_1, \dots, p_n\}$$

matches a set that has n elements e_1, \dots, e_n such that p_i matches e_i and $e_i < e_j$ for $i < j$. The pattern

$$\{p_1, \dots, p_n, \delta\}$$

matches a set that has $m \geq n$ elements such that its first n elements match the patterns p_i , and the set consisting of the other $m - n$ elements matches the δ -pattern.

Similarly, the pattern

$$\{p_1 \rightarrow q_1, \dots, p_n \rightarrow q_n\}$$

matches a map consisting of n key/value pairs such that the key/value pairs match the pattern pairs in order. The pattern $\{ \rightarrow \}$ matches the empty map. Map patterns can have a δ pattern, too.

18. REALS AND INTERVAL ARITHMETIC

Babel-17 is radical in its treatment of floating point arithmetic: there is *only* interval arithmetic. What that means is that reals are represented in Babel-17 as closed real intervals, i.e. as pairs $[a; b]$ where a and b are floating point numbers and $a \leq b$. The usual floating point numbers can then be viewed as that subset of reals for which $a = b$ holds. For example, the notation 1.2 is just short for $[1.2; 1.2]$.

The general form of this interval notation is

$$[a; b]$$

Its value is formed by evaluating a and b , resulting in **reals** $[a_1; a_2]$ and $[b_1; b_2]$, respectively, and then forming the interval $[a_1; b_2]$ if $a_2 \leq b_1$. In case the evaluation does not yield reals, a *DomainError*-exception is thrown. In case $a_2 > b_1$, an *EmptyReal*-exception is raised.

The general rule of interval arithmetic in Babel-17 is as follows: If f is a mapping from \mathbb{R}^n to \mathbb{R} , then for its interval arithmetic implementation F the following must hold for all $x_i \in \mathbb{R}$ and all **reals** $y_i = [u_i; v_i]$ such that $u_i \leq x_i \leq v_i$:

$$f(x_1, \dots, x_n) \in F(y_1, \dots, y_n)$$

The built-in order on **reals** is defined such that

$$[a_1; a_2] \leq [b_1; b_2] \quad \text{iff} \quad a_2 < b_1 \vee (a_1 = b_1 \wedge a_2 = b_2)$$

Note that certain seemingly obvious inequalities like $[1.0; 2.0] \leq 2.0$ do *not* hold! Also note that while $[1.0; 2.0] < 3.0$ is true, the seemingly identical inequality $[1.0; 2.0] < 3$ is false! This is because *Integers* and *Reals* are unrelated with respect to the built-in order.

19. SYNTACTIC SUGAR

One of the goals of Babel-17 is that Babel-17 code is easy to read and understand. I have found that allowing arbitrary user-specific extensions to the syntax of code is definitely not helping to achieve this goal. Nevertheless, a bare minimum of syntactic sugar is necessary to support the most basic conventions known from arithmetic; for example, one should be able to write $3 + 5$ to denote the addition of the integer 3 and the integer 5.

The programmer can use some of this syntactic sugar when defining her own objects. For example, $3 + 5$ is just syntactic sugar for

$$3.\text{plus_} 5$$

TABLE 8. Syntactic Sugar

Sugared	Desugared
$a + b$	$a.\text{plus_} b$
$a - b$	$a.\text{minus_} b$
$- a$	$a.\text{uminus_}$
$a * b$	$a.\text{times_} b$
a / b	$a.\text{slash_} b$
$a \text{ div } b$	$a.\text{div_} b$
$a \text{ mod } b$	$a.\text{mod_} b$
$a \wedge b$	$a.\text{pow_} b$
$a ++ b$	$a.\text{plus_}_ b$
$a -- b$	$a.\text{minus_}_ b$
$a ** b$	$a.\text{times_}_ b$
$a // b$	$a.\text{slash_}_ b$
$a \text{ to } b$	$a.\text{to_} b$
$a \text{ downto } b$	$a.\text{downto_} b$
$f \ x$	$f.\text{apply_} x$

Table 8 lists all syntactic sugar of Babel-17 that is also available to the programmer. The availability of syntactic sugar for function application means that you can let your own objects behave as if they were functions.

20. MEMOIZATION

Babel-17 supports *memoization*. In those places where **def**-statements can be used, **memoize**-statements can be used, also. A **memoize**-statement must always refer to **def**-statements in the same scope. Babel-17 differentiates two kinds of memoization, *strong* and *weak*. A **memoize**-statement has the following syntax:

memoize $ref_1 \dots ref_n$

The ref_i are either of the form *id* to indicate strong memoization, or of the form (*id*) to signal weak memoization. In both cases *id* refers to the **def**-statement being memoized. As an example, here is the definition of the weakly memoized fibonacci-sequence:

```
memoize (fib)
def fib 0 = 0
def fib 1 = 1
def fib n = fib (n-1) + fib (n-2)
```

The difference between weak and strong memoization is that strong memoization always remembers a value once it has been computed; weak memoization instead may choose not to remember computed results in certain situations, for example in order to free memory in low memory situations. Note that all arguments to the same memoized function must be totally ordered by the built-in partial order.

21. LINEAR SCOPE

We are now ready to explore the full power of block expressions and statements in Babel-17. We approach the topic of this and the following sections, *linear scope*, in an example-oriented way. A more in-depth treatment can be found in my paper *Purely Functional Structured Programming*, also available at www.babel-17.com.

First, we extend the syntax of **val**-statements such that it is legal to leave out the keyword **val** in certain situations. Whenever the situation is such that it is legal to write

$$a = \text{expr}$$

for a variable identifier a , we say that a is in *linear scope*. We call the above kind of statements *assignments*. A necessary condition for a to be in linear scope is for a to be in scope because of a previous **val**-statement or a pattern match. The idea behind linear scope is that when the program control flow has a linear structure, then we can make assignments to variables in linear scope without leaving the realm of purely functional programming.

The following two expressions are equivalent:

begin val x = a val y = b val x = c d end	begin val x = a val y = b x = c d end
---	--

So far, **val**-statements and assignments have indistinguishable semantics. The differences start to show when we look at nested block expressions:

begin val x = 1 val y = 2 begin val x = 3 val y = 4 * x end (x, y) end	begin val x = 1 val y = 2 begin val x = 3 y = 4 * x end (x, y) end	begin val x = 1 val y = 2 begin val x = 3 val y = 0 y = 4 * x end (x, y) end
evaluates to (1, 2)	evaluates to (1, 12)	evaluates to (1, 2)

The above examples show that the effect of dropping the **val** in a **val** statement is that the binding of an identifier becomes visible at that level within the linear scope where it has last been introduced via a **val** statement or via a pattern match.

Linear scope spreads along the statements and nested block expressions of a block expression. It usually does not spread into expressions. An exception are certain expressions that can also be viewed as statements. We call these expressions *control expressions*. For example, in

```
begin
  val x = 1
```

```

    x = 2
    val y = a * b
    (x, y)
end

```

the linear scope of x does not extend into the expressions a and b , because $a * b$ is not a control expression. Therefore the above expression will always evaluate (assuming there is no exception) to a pair which has 2 as its first element. But for example in

```

begin
  val x = 1
  x = 2
  val y =
    begin
      s1
      ⋮
      sn
    end
  (x, y)
end

```

the linear scope of x extends into s_1 and spreads then along the following statements. Therefore the first element of the pair that is the result of evaluating above expression depends on what happens in the s_i . Here are three example code snippets that further illustrate linear scope:

<pre> begin val x = 1 val y = begin x = 2 x+x end (x, y) end </pre>	<pre> begin val x = 1 val y = 3 * begin x = 2 x+x end (x, y) end </pre>	<pre> begin val x = 1 val y = 3 * begin val x = 2 x+x end (x, y) end </pre>
evaluates to (2, 4)	illegal	evaluates to (1, 12)

These are the control expressions that exist in Babel-17:

- **begin ... end**
- **if ... end**
- **match ... end**
- **for ... end**
- **while ... end**

The last two control expressions are loops and explained in the next section. The first three have already been treated without delving too much into their statement character. We have already seen how **begin ... end** is responsible for nesting block expressions, when it is used as a statement. Just as the **begin ... end** expression may be used as a statement, you can also use all other control expressions as statements, for example:

```

begin
  val x = random 2
  if x == 0 then
    x = 100
  else
    x = 200
  end
  x + x
end

```

This expression will evaluate either to 200 or to 400.

For **if**-statements the **else**-branch is optional, but **match**-statements throw a `NoMatch` exception if none of the patterns matches.

22. RECORD UPDATES

Linear scope makes it possible to have purely functional record updates in Babel-17. Let us assume you have defined u via

```
val u = {x = 10, y = 20, z = -4}
```

and now you want to bind u to another record that differs only in the x -component. You could proceed as follows:

```
u = {x = 9, y = u.y, z = u.z}
```

but this clearly does not scale with the number of components of u . A more scalable alternative would be to write

```

u =
  object + [u]
  def x = 9
end

```

Babel-17 allows to write down the above statement in a more concise form:

```
u.x = 9
```

In general,

```
u.m = t
```

is shorthand notation for

```

u =
  begin
    val evaluated_t = t
    object + [u]
    def m = evaluated_t
  end
end

```

For the sake of completeness, there is also the much less useful notation

```
val u.m = t
```

which is short for

```
val u = { m = t }
```

23. WITH

By default, all yielded values of a block expression are collected into a list which collapses in the case of a single element. The programmer might want to deviate from this default and collect the yielded values differently, for example to get rid of the collapsing behavior. The **with** expression allows her to do just that. Its syntax is:

```
with c do
  b
end
```

where c is a *collector* and b is a block expression. A collector c is any object that

- responds to the message `collector_close_`,
- and returns via `c.collector_add_ x` another collector.

It is recommended that collectors also support the message **empty** to represent the empty collector that has not collected anything yet.

Lists, vectors, sets, maps and strings are built-in collectors which the programmer can use out-of-the-box; apart from that she can implement her own collectors, of course.

Here is an example where we use a set as a collector:

```
with {4} do
  yield 1
  yield 2
  yield 1
  10
end
```

Above expression evaluates to {1, 2, 4, 10}.

24. LOOPS

This is the syntax for the **while**-loop:

```
while c do
  b
end
```

Here c must evaluate to a boolean and b is a block expression. For example, here is how you could code the euclidean algorithm for calculating the greatest common divisor:

```
def gcd (a,b) = begin
  while b != 0 do
    (a, b) = (b, a mod b)
  end
  a
end
```

There is also the **for**-loop. It has the following syntax:


```

for  $p$  in  $C$  do
   $b$ 
end

```

In the above p is a pattern, C is a *collection*, and b is a block expression. The idea is that above expression iterates through those elements of the *collection* C which match p ; for each successfully matched element, b is executed.

An object C is a collection if it handles the message `iterate_`

- by returning `()` if it represents the empty collection,
- or otherwise by returning (e, C') such that C' is also a collection.

Here is an example of a simple **for**-loop expression:

```

begin
  val  $s = [10, (5, 8), 7, (3,5)]$ 
  with  $\{->\} : \text{for } (a,b) \text{ in } s \text{ do}$ 
    yield  $(b,a)$ 
  end
end

```

evaluates to $\{8 -> 5, 5 -> 3\}$.

Using **for**-loops in combination with linear scope it is possible to formulate all of those *fold*-related functionals known from functional programming in a way which is easier to parse (and remember) for most people. Let us for example look at a function that takes a list m of integers $[a_0, \dots, a_n]$ and an integer x as arguments and returns the list

$$[q_0, \dots, q_n] \quad \text{where} \quad q_k = \sum_{i=0}^k a_i x^i$$

The implementation in Babel-17 via a loop is straightforward, efficient and even elegant:

```

 $m ==> x ==>$ 
  with  $[]$  do
    val  $y = 0$ 
    val  $p = 1$ 
    for  $a$  in  $m$  do
       $y = y + a*p$ 
       $p = p * x$ 
      yield  $y$ 
    end
  end

```

The built-in collections that can be used with **for**-loops are the usual suspects: lists, vectors, sets, maps and strings. Of course you can define your own custom collections. There is even a pattern you can use for matching against an arbitrary collection:

```

(for  $p_1, \dots, p_n$  end)

```

and the corresponding δ pattern

```

(for  $p_1, \dots, p_n, \delta$  end)

```

match collections with exactly n or at least n elements, respectively.

25. MODULES AND TYPES

26. PRAGMAS

Pragmas are statements that are not really part of the program, but inserted in it for pragmatical reasons. They are useful for testing, debugging and profiling a program. There are currently four different kinds of pragmas available:

#log *e* evaluates the expression *e* and logs it.

#print *e* evaluates the expression *e* such that it has no internal lazy or concurrent computations any more, and then logs it.

#profile *e* evaluates the expression *e*, gathers profiling information while doing so, and logs both.

#assert *e* evaluates the expression *e* and signals an error if *e* does not evaluate to **true**.

The Babel-17 interpreter / compiler is free to ignore pragmas, typically if instructed so by the user.

27. UNIT TESTS

28. STANDARD LIBRARY (TO BE UPDATED FOR V0.3)

The standard library of Babel-17 is just the description of all messages that the built-in types of Babel-17 respond to. The majority of these messages have special syntactic support by the language. The standard library is therefore an important part of the language definition of Babel-17.

The messages of the standard library can be grouped as follows:

- collector related messages (Section 23),
- collection related messages (Section 24),
- messages that convert between different types,
- those messages that are listed as syntactic sugar in Table 8,
- messages specific to the type.

As a convention, all messages that are normally only used via some sort of syntactic sugar end with an underscore in their name.

28.1. Collections and Collectors. The built-in types that can be used as collectors and collections are: List, Vector, Set, Map and String. In the case of maps, the elements of the collection are pairs (vectors of length 2) where the first element of the pair represents the key, and the second element the value the key is mapped to. For strings, the elements of the collection are strings of length 1, consisting of a single Unicode code point.

All built-in collection/collector types implement the messages described in Table 9.

28.2. Type Conversions. For many built-in types, there is a message that converts a value into the corresponding value of this type. Not all values can be converted to every type. If a value cannot fulfill a conversion request, it throws an exception with parameter `DomainError`. Table 10 lists all type conversion messages and those source types such that at least some of their values can be successfully converted via the given message. Note that Babel-17 does no automatic type conversion except for the conversion between

TABLE 9. Additional messages for collection/collector c

Message	Description
$c.\text{isEmpty}$	checks if c is an empty collection
$c.\text{empty}$	an empty collection that has the same type as c
$c.\text{size}$	the size of collection c
$c + x$	adds x as member to collection c
$c ++ d$	adds all members of d as members to c
$c - x$	removes from c all members that are equal to x
$c -- d$	removes from c all members that are equal to an element in d
$c ** d$	removes from c all members that are not equal to an element in d
$c.\text{head}$	the first element of c
$c.\text{tail}$	all elements of c except the first one
$c.\text{atIndex } i$	the i -th element of c
$c.\text{indexOf } x$	the lowest i such that $c.\text{atIndex } i == x$
$c.\text{contains } x$	checks if c contains x
$c.\text{take } n$	forms a collection out of the first n elements of c
$c.\text{drop } n$	forms a collection by dropping the first n elements of c
c/f	maps the function f over all elements of c
$(c * f) a_0$	folds f over c via $a_{i+1} = f(c_i, a_i)$
$c \hat{ } f$	filters c by boolean-valued function f
$c // f$	map created by all key/value pairs $(x, f(x))$ where x runs through the elements of c ; later pairs overwrite earlier pairs

TABLE 10. Type Conversions

Message	Source Types
integer	Boolean, String
infinity	String
boolean	Integer, String
string	Integer, Infinity, Boolean, List, Vector
list	Vector, Set, Map, String
vector	List, Set, Map, String
set	List, Vector, Map
map	List, Vector, Set

lists and vectors.

In the rest of this section we enumerate for each built-in type all messages that are supported by this type. Collector, collection and type conversion messages are excluded from this enumeration.

TABLE 11. List/Vector-specific messages

$l\ i$	same as $l.\text{atIndex } i$
$-l$	reverses l

TABLE 12. Map-specific Messages

$c.\text{contains } x$	checks if c contains x
$c.\text{containsKey } k$	checks if c contains (k, v) for some v
$m + (k, v)$	map created from the map m by associating k with v
$m - k$	map created from the map m by removing the key k
$m ++ n$	map created from the map m by adding the key/value pairs that are elements of n
$m -- n$	map created from m by removing all keys that are elements of n
$m ** n$	map created from m by removing all keys that are not elements of n
$m\ k$	returns the value v associated with k in m , or returns a dynamic exception with parameter <code>DomainError</code> if no such value exists
$m\ //\ f$	map created by applying the function f to the key/value pairs (k, v) of m , yielding key/value pairs $(k, f(k, v))$

28.3. Integer, Infinity. Integers can be arbitrarily large and support the usual operations (+, binary and unary -, *, ^, **div**, **mod**). Division and modulo are Euclidean. Where it makes sense, infinite values can be used.

The expression $a\ \text{to } b$ denotes the list of values $a, a + 1, \dots, b$. The expression $a\ \text{downto } b$ denotes the list of values $a, a - 1, \dots, b$.

28.4. String. Implements no messages specific for strings. The semantics of the `indexOf` and `contains` messages is extended with respect to the usual collection semantics to not only search for strings of length 1, but arbitrarily long strings.

28.5. List and Vector. Those messages specific for lists and vectors are listed in Table 11.

28.6. Set. There are no messages specific for sets except that for a set s the expression $s\ x$ is equivalent to $s.\text{contains } x$ and therefore tests if x is an element of s .

28.7. Map. The messages specific to maps are listed in Table 12. Note that the operators -, --, ** and // deviate in their semantics from the usual collection semantics, but that the operators + and ++ *do* behave according to the usual collection semantics.

28.8. Object. Custom objects respond exactly to the set of messages defined in their body, with one exception: if the custom object implements all four of these messages:

- `collector_add_`
- `collector_close_`
- `empty`
- `collector_iterate_`

then the custom object automatically inherits standard implementations for all messages listed in Table 9 and for the three type conversion messages `list`, `vector` and `set`. These standard implementations are shadowed by actual implementations the custom object might provide.

29. WHAT'S NEXT?

Where will Babel-17 go from here? There are two dimensions along which Babel-17 has to grow in order to find adoption.

Maturity and breadth of implementations While there will probably always be a reference implementation of Babel-17 that strives for simplicity and clarity and sacrifices performance to achieve this, Babel-17 will need quality implementations on major computing platforms like JavaVM, Android, iOS and HTML 5. A lot of the infrastructure of these implementations can be shared, so once there is a speedy implementation of Babel-17 running for example on the Java Virtual Machine, the other implementations should follow more quickly. For example, there are many many opportunities for both static and particularly dynamic optimizations of the standard library implementation.

Development of the language Right now Babel-17 is a purely functional programming language with no static typing and only a handful of built-in types. Modularization and limited encapsulation of code is already possible via objects. More powerful encapsulation, more powerful types and some way of dealing with state, user interfaces, and communication with the outside world will be added to future versions of Babel-17. Programmers will be able to choose which of these future features they would like to employ. Those they don't care about they will be able to just ignore mostly and pretend that Babel-17 v21.0 is still Babel-17 v0.21.