

# SYNTAX AND SEMANTICS OF BABEL-17, V0.21 (DRAFT)

STEVEN OBUA

## CONTENTS

1. Introduction	1
2. Lexical Matters	2
3. Overview of Built-in Types	3
4. Objects and Functions	3
5. Exceptions	5
6. Laziness and Concurrency	5
7. Lists and Vectors	6
8. CExprs	6
9. Pattern Matching	7
10. Non-Determinism in Babel-17	7
11. Block Expressions	9
12. Anonymous Functions	11
13. Objects	12
14. Boolean Operators	13
15. Order	14
16. Sets and Maps	15
17. Patterns for Sets and Maps	17
18. API and Runtime Environment	17
19. Syntactic Sugar	17
20. Memoization	17
21. Linear Scope	18
22. Yield	20
23. Loops	21

## 1. INTRODUCTION

The first question that someone who creates a new programming language will hear from others inevitably is: Why another programming language? Are there not already enough programming languages out there that you can pick from?

I always wanted a programming language

- in which I can express myself compactly and without jumping through too many hoops,
- that is easy to understand and to learn,

---

*Date:* October 10, 2010.

- in which purely functional programming is the default and not the exception,
- that supports exceptions,
- that has pattern matching,
- that has built-in support for laziness and concurrency,
- that has lists, vectors, sets and maps,
- that seamlessly marries structured programming constructs like loops with purely functional programming,
- that supports objects, modules and data encapsulation,
- that has mature implementations for all major modern computing platforms,
- that is beautiful.

There is no such language out there. Therefore I decided to create one. When Babel-17 will have reached v1.0 then the above goals will have been achieved. The version described in this document is Babel-17 v0.21, so there is still some distance to go.

Babel-17 is not a radically new or revolutionary programming language. It picks those raisins I like out of languages like Standard ML, Alice ML, Java, Javascript, Scala, Erlang, Haskell, Lisp, Clojure and Mathematica, and tries to evolve them into a beautiful new design. Of course, what some regard as beautiful, others might find unremarkable, or even ugly.

Babel-17 is a functional language. It is not a pure functional language because it contains sources of non-determinism like the ability to pick one of two values randomly. If you take these sources of non-determinism out of the language, it becomes pure.

In Babel-17, every value has a type. These types are not checked statically, but only dynamically.

Babel-17 is also object-oriented in that many aspects of Babel-17 can be explained in terms of sending messages to objects.

The default evaluation order of Babel-17 is strict. But optional laziness is built into the heart of Babel-17, also, and so is concurrency.

Babel-17 has pattern matching and exception handling. Both are carefully integrated with each other and with the laziness and concurrency capabilities of Babel-17.

Furthermore, Babel-17 establishes the paradigm of *purely functional structured programming* via the notion of linear scope.

In this document I specify Babel-17 in a mostly informal style: formal notation is sometimes used when it supports clarity.

## 2. LEXICAL MATTERS

Babel-17 source code is always written UTF-8 encoded. If it ain't UTF-8, it ain't Babel-17.

Constructors are alphanumeric strings which can also contain underscores, and which must start with a capital letter. Identifiers are alphanumeric strings which can also contain underscores, and which must start with a non-capital letter. The distinction between two constructors or two identifiers is not sensitive to capitalization.

These are the keywords of Babel-17:

```

begin  end  object  with  if  then  else  elseif
while  for  do  await  random  yield  match  case
as  val  def  in  exception  lazy  concurrent  memoize
to  downto  true  false  this  infinity  nil

```

Note that keywords are written always in lower case. For example, `bEGIN` is not a legal keyword, but it also isn't an identifier (because of case insensitivity). Also note that `Begin` and `BEGIN` denote the same legal constructor. When we talk about identifiers in the rest of the paper, we always mean identifiers which are not keywords.

You can write down decimal, hexadecimal, binary and octal numbers in Babel-17. The following notations all denote the same number:

```
15  0xF  0b1111  0o17
```

Babel-17 also knows the number **infinity**.

Strings start and end with a quotation mark. Between the quotation mark, any valid UTF-8 string is allowed that does not contain newline, backslash or quotation mark characters, for example:

```
"hello world" "$1, 2%" "1 <= 4! <= 5^2"
```

Newline, backslash and quotation mark characters can be written down in escaped form. Actually, any unicode character can be written down in its escaped form via the hexadecimal value of its codepoint:

String consisting only of a quotation mark	"\""
String consisting only of a backslash	"\\"
String consisting only of a line feed	"\n"
String consisting only of a carriage return	"\r"
String consisting only of line feed via 16-bit escaping	"\u000A"
String consisting only of line feed via 32-bit escaping	"\U0000000A"

Comments in Babel-17 obey the same syntax as in C: There are multiline comments written between matching pairs of `/*` and `*/`, and there are comments that occupy the rest of the line which start with a `//`.

Finally, here are all ASCII special symbols that can occur in a Babel-17 program:

```

==  !=  <>  <  <=  >  >=  +  -  *  /  %  ^
|  &  !  ++  --  **  ;  ,  ::  ->  =>  ?  ...
(  )  [  ]  {  }  .  =

```

Table 1 lists all other Unicode symbols that have a meaning in Babel-17.

### 3. OVERVIEW OF BUILT-IN TYPES

Each value in Babel-17 has a unique type. All built-in types are depicted in Table 2.

### 4. OBJECTS AND FUNCTIONS

All values in Babel-17 are objects. This means that you can send messages to them. Many values in Babel-17 are functions. This means that you can apply them to an argument, yielding another value. Note that for a value to be an object, it does not need to have type *Object*, and for a value to be a function, it does not need to have type *Function*.

TABLE 1. Further Unicode Symbols

Unicode Hex Code	Display	ASCII Equivalent
2261	$\equiv$	<code>==</code>
2262	$\neq$	<code>!=</code>
2264	$\leq$	<code>&lt;=</code>
2265	$\geq$	<code>&gt;=</code>
221E	$\infty$	<code>infinity</code>
2227	$\wedge$	<code>&amp;</code>
2228	$\vee$	<code> </code>
00AC	$\neg$	<code>!</code>
2237	$::$	<code>::</code>
2192	$\rightarrow$	<code>-&gt;</code>
21D2	$\Rightarrow$	<code>=&gt;</code>
2026	$\dots$	<code>...</code>

TABLE 2. Built-in Types of Babel-17

Type	Description
<i>Integer</i>	the type of arbitrary size integer numbers
<i>Infinity</i>	the type consisting of the values <b>infinity</b> and <b>-infinity</b>
<i>Boolean</i>	the type consisting of the values <b>true</b> and <b>false</b>
<i>String</i>	the type of valid Unicode strings
<i>List</i>	the type of lists
<i>Vector</i>	the type of vectors / tuples
<i>Set</i>	the type of sets
<i>Map</i>	the type of maps
<i>CExpr</i>	the type of constructed expressions
<i>Object</i>	the type of user-defined objects
<i>Function</i>	the type of functions
<i>PersistentException</i>	the type of persistent exceptions
<i>DynamicException</i>	the type of dynamic exceptions

The result of sending message  $m$  to object  $v$  is written as

$$v.m$$

Here  $m$  is an identifier. The result of applying function  $f$  to value  $x$  is written as

$$f\ x$$

Note that  $f\ x$  is equivalent to

$$(f.\text{apply})\ x$$

Therefore any object that responds to the apply message can act as a function. In the above we could also leave out the brackets because sending messages binds stronger than function application.

Repeated function application associates to the left, therefore

$$f\ x\ y$$

is equivalent to  $(fx) y$ .

## 5. EXCEPTIONS

Exception handling in Babel-17 mimicks exception handling mechanisms like those which can be found in Java or Standard ML or Erlang, while adhering at the same time to the functional paradigm.

There are two types of exceptions: *PersistentException* and *DynamicException*. The difference is that a *PersistentException* can be treated as part of any data structure, and is passed around just like any other value, while a *DynamicException* can never be part of a data structure and has special evaluation semantics.

Exceptions in Babel-17 are not that special at all but mostly just another type of value. Let us write *PersistentException v* for a *PersistentException* with parameter  $v$ , and *DynamicException v* for a *DynamicException* with parameter  $v$ . We also write *Exception v* for an exception with parameter  $v$ . The parameter  $v$  is a non-exceptional value; this means that  $v$  can be any Babel-17 value except one of type *DynamicException*. Note that  $v$  can be non-exceptional and of type *PersistentException* at the same time.

Then the following deterministic rules govern how exceptions behave with respect to sending of messages, function application, laziness and concurrency:

$$\begin{array}{ll}
 (\text{Exception } v).m & \rightsquigarrow \text{DynamicException } v \\
 (\text{Exception } v) x & \rightsquigarrow \text{DynamicException } v \\
 f (\text{DynamicException } v) & \rightsquigarrow \text{DynamicException } v \quad \text{where } f \text{ is non-exceptional} \\
 \text{exception } v & \rightsquigarrow \text{DynamicException } v \\
 \text{lazy } (\text{Exception } v) & \rightsquigarrow \text{PersistentException } v \\
 \text{concurrent } (\text{Exception } v) & \rightsquigarrow \text{PersistentException } v
 \end{array}$$

Exceptions in Babel-17 are created with the expression **exception**  $v$ . Creating exceptions in Babel-17 corresponds to *raising* or *throwing* exceptions in other languages. There is no special construct for catching exceptions in Babel-17. Instead, catching an exception can be done via the **match** expression which will be described later.

In a later section, we will describe the **lazy** and **concurrent** expressions of Babel-17. They are the reason why exceptions are divided into dynamic and persistent ones.

## 6. LAZINESS AND CONCURRENCY

The default evaluation mechanism of Babel-17 is strict. This basically means that the arguments of a function are evaluated before the function is applied to them. Babel-17 has two constructs to change this default behaviour.

The expression

**lazy**  $e$

is not evaluated until it is actually needed. When it is needed, it evaluates to whatever  $e$  evaluates to, with the exception of dynamic exceptions, which are converted to persistent ones. It is guaranteed that  $e$  is only evaluated once.

The expression

**concurrent**  $e$

also evaluates to whatever  $e$  evaluates to, again with the exception of dynamic exceptions which are converted to persistent ones. This evaluation will happen concurrently. It is guaranteed that  $e$  is only evaluated once.

One could think that apart from obvious different performance characteristics, the expressions **lazy**  $e$  and **concurrent**  $e$  are equivalent. This is not so. If  $e$  is a non-terminating expression then, even if the value of **concurrent**  $e$  is never needed during program execution, it might still lead to a non-terminating program execution. In other words, the behaviour of **concurrent**  $e$  is unspecified for non-terminating  $e$ .

## 7. LISTS AND VECTORS

For  $n \geq 0$ , the expression

$$[e_1, \dots, e_n]$$

denotes a *list* of  $n$  elements.

The expression

$$(e_1, \dots, e_n)$$

denotes a *vector* of  $n$  elements, at least for  $n \neq 1$ . For  $n = 1$ , there is a problem with notation, though, because  $(e)$  is equivalent to  $e$ . Therefore there is the special notation  $(e,)$  for vectors which consist of only one element.

The only difference between lists and vectors is that they have different performance characteristics for the possible operations on them. Lists behave like simply linked lists, and vectors behave like arrays. Note that all data structures in Babel-17 are immutable. Apart from performance characteristics, anywhere you can use a list, you can use a vector, and vice versa.

Another way of writing down lists is via the right-associative  $::$  constructor:

$$h::t$$

Here  $h$  denotes the *head* of the list and  $t$  its *tail*. Actually, note that the expression  $[e_1, \dots, e_n]$  is just syntactic sugar for the expression

$$e_1::e_2::\dots::e_n::[]$$

Dynamic exceptions cannot be part of a list but are propagated:

$$\begin{aligned} (\text{DynamicException } v)::t &\rightsquigarrow \text{DynamicException } v \\ h::(\text{DynamicException } v) &\rightsquigarrow \text{DynamicException } v \quad \text{where } h \text{ is non-exceptional} \end{aligned}$$

## 8. CEXPRs

A *CExpr* is a constructor  $c$  together with a parameter  $p$ , written  $c \ p$ . It is allowed to leave out the parameter  $p$ , which then defaults to  $()$ . For example, HELLO is equivalent to HELLO  $()$ . A constructor  $c$  cannot have a dynamic exception as its parameter, therefore we have:

$$c \ (\text{DynamicException } v) \rightsquigarrow \text{DynamicException } v$$

## 9. PATTERN MATCHING

Maybe the most powerful tool in Babel-17 is pattern matching. You use it in several places, most prominently in the **match** expression which has the following syntax:

```
match  $e$ 
  case  $p_1 \Rightarrow b_1$ 
   $\vdots$ 
  case  $p_n \Rightarrow b_n$ 
end
```

Given a value  $e$ , Babel-17 tries to match it to the patterns  $p_1, p_2, \dots$  and so on sequentially in that order. If  $p_i$  is the first pattern to match, then the result of **match** is given by the block expression  $b_i$ . If none of the pattern matches then there are two possible outcomes:

- (1) If  $e$  is a dynamic exception, then the value of the match is just  $e$ .
- (2) Otherwise the result is a dynamic exception with parameter NoMatch  $e$ .

A few of the pattern constructs are capable of raising dynamic exceptions because they incorporate arbitrary value expressions. When during a pattern match such an exception occurs, then the entire match is aborted and the dynamic exception is returned as the result of the **match**-expression. Furthermore note that persistent exceptions in  $e$  or any of the patterns also lead to a dynamic exception if the value of the persistent exception is forced for pattern matching.

So what does a pattern look like? Table 3 lists all ways of building a pattern.

In this table of pattern constructions we use the  $\delta$ -pattern  $\delta$ . This pattern stands for "the rest of the collection under consideration" and can be constructed by the following rules:

- (1) The ellipsis  $\dots$  is a  $\delta$ -pattern that matches any collection.
- (2) If  $\delta$  is a  $\delta$ -pattern, and  $x$  an identifier, then  $(x \text{ as } \delta)$  is a  $\delta$ -pattern.
- (3) If  $\delta$  is a  $\delta$ -pattern, and  $e$  an expression, then  $(\delta \text{ if } e)$  is a  $\delta$ -pattern.
- (4) If  $\delta$  is a  $\delta$ -pattern, and  $e$  an expression, then  $(e ? \delta)$  is a  $\delta$ -pattern.

Note that pattern matching does not distinguish between vectors and lists. A pattern that looks like a vector can match a list, and vice versa.

## 10. NON-DETERMINISM IN BABEL-17

One source of non-determinism in Babel-17 is *probabilistic* non-determinism. The expression

```
random  $n$ 
```

returns for an integer  $n > 0$  a number between 0 and  $n - 1$  such that each number between 0 and  $n - 1$  has equal chance of being returned. If  $n$  is a dynamic exception, then this exception is propagated, if  $n$  is a non-exceptional value that is not an integer  $> 0$  then the result is an exception with parameter InvalidArgument  $n$ .

The other source of non-determinism is *concurrent* non-determinism. The expression

```
await  $l$ 
```

TABLE 3. Pattern Constructs in Babel-17

Syntax	General Description
-	the underscore symbol matches anything but a dynamic exception
$x$	an identifier $x$ matches anything but a dynamic exception and binds the matched expression to $x$
$(x \text{ as } p)$	matches $p$ , and binds the successfully matched value to $x$ ; the match fails if $p$ does not match or if the matched value is a dynamic exception
$z$	a number $z$ , like 0 or 42 or $-10$ or <b>infinity</b> or <b>-infinity</b> , matches just that number $z$
$c \ p$	matches a <i>CExpr</i> with constructor $c$ if the parameter of the <i>CExpr</i> matches $p$
$s$	a string $s$ , like "hello", matches just that string $s$
$[p_1, \dots, p_n]$	matches a list/vector consisting of $n \geq 0$ elements, such that element $e_i$ of the list is matched by pattern $p_i$
$(p_1, \dots, p_n)$	matches a vector/list consisting of $n = 0$ or $n \geq 2$ elements, such that element $e_i$ of the list is matched by pattern $p_i$
$(p)$	same as $p$
$[p_1, \dots, p_n, \delta]$	matches a list/vector consisting of at least $n \geq 1$ elements, such that the first $n$ elements $e_i$ of the list/vector are matched by the patterns $p_i$
$(p_1, \dots, p_n, \delta)$	matches a vector/list consisting of at least $n \geq 1$ elements, such that the first $n$ elements $e_i$ of the vector/list are matched by the patterns $p_i$
$(p, )$	matches a vector/list consisting of a single element that matches the pattern $p$ .
$(h::t)$	matches a non-empty list/vector such that $h$ matches the head of the list/vector and $t$ its tail
$(p \text{ if } e)$	matches $p$ , but only if $e$ evaluates to <b>true</b> ; identifiers bound in $p$ can be used in $e$
$(\text{val } e)$	matches any value which is equivalent to $e$
$(f ? p)$	here $f$ is a function; $f$ is applied to the value to be matched; the match succeeds if the result of the application is (Some $r$ ) and $p$ matches $r$ , and fails if the result is None; otherwise the dynamic exception <code>TypeError</code> is thrown
$(d ? p)$	here $d$ is a not a function and non-exceptional; short for $((v \Rightarrow v.\text{deconstruct } d) ? p)$
$(e ?)$	similar as the two patterns above but tests for <b>true</b> / <b>false</b> instead
$\{p_1, \dots, p_n\}$	see section 17
$\{p_1, \dots, p_n, \delta\}$	see section 17
$\{q_1 \rightarrow p_1, \dots, q_n \rightarrow p_n\}$	see section 17
$\{q_1 \rightarrow p_1, \dots, q_n \rightarrow p_n, \delta\}$	see section 17
$\{->\}$	matches the empty map
$\{m_1 = p_1, \dots, m_n = p_n\}$	matches an object that has exactly the messages $m_1, \dots, m_n$
$\{m_1 = p_1, \dots, m_n = p_n, \delta\}$	matches an object that has the messages $m_1, \dots, m_n$
<b>nil</b>	matches the empty object
<b>exception</b> $p$	matches any exception such that its parameter matches $p$



takes a finite list  $l$  of possibly concurrently evaluated expressions and waits until at least one of the element expressions has finished computing. It returns a list of the element expressions that have been computed so far. Again, in case  $l$  is non-exceptional and not a finite list or vector, the result is an exception with parameter `InvalidArgument`  $l$ . If  $l$  is a dynamic exception, then it is propagated such that the result is  $l$ .

## 11. BLOCK EXPRESSIONS

So far we have only looked at expressions. We briefly mentioned the term *block expressions* in the description of the **match** function, though. We will now introduce and explain block expressions.

Block expressions can be used in several places as defined by the Babel-17 grammar. For example, they can be used in a **match** expression to define the value that corresponds to a certain case. But block expressions can really be used just everywhere where a normal expression is allowed:

```
begin
  b
end
```

is a normal expression where  $b$  is a block expression. A block expression has the form

```
s1
:
sn
e
```

where the  $s_i$  are statements and  $e$  is the return expression of the block (a return expression is just a normal expression). In a block expression newlines can be used to separate the statements and the return expression from each other. The other means of separation are semicolons. Later on we will see how block expressions can be used to imitate imperative programming; on those occasions there will be no return expression  $e$  but only statements.

Statements are Babel-17's primary tool for introducing identifiers. There are several kinds of statements. Two of them will be introduced in this section.

First, there is the **val**-statement which has the following syntax:

```
val p = e
```

Here  $p$  is a pattern and  $e$  is an expression. Its meaning is that  $e$  gets evaluated and then is matched to  $p$ . If the match is successful then all identifiers bound by the match can be used in later statements and in the return expression of the block expression. Otherwise, some kind of dynamic exception must have happened which then becomes the value of the entire block expression.

Second, there is the **def**-statement which obeys the following syntax for defining the identifier  $id$ :

```
def id arg = e
```

The  $arg$  part is optional. If  $arg$  is present, then it must be a pattern. Let us call those definitions where  $arg$  is present a *function definition*, and those definitions where  $arg$  is not present a *simple definition*.

TABLE 4. Legal and illegal definitions

<b>val</b> x = y	<b>def</b> x = y	<b>val</b> x = y	<b>def</b> x = y
<b>val</b> y = 0	<b>val</b> y = 0	<b>def</b> y = 0	<b>def</b> y = 0
<i>illegal</i>	<i>illegal</i>	<i>legal</i>	<i>legal</i>

Per block expression and identifier *id* there can be either a single simple definition, or there can be several function definitions. If there are multiple function definitions for the same identifier, then they must appear sequentially in the block expression, without any other statements interrupting the sequence.

The defining expressions in **def**-statements can recursively refer to the other identifiers defined by **def**-statements in the same block expression. This is the main difference between definitions via **def** and definitions via **val**. Only those **val**-identifiers are in scope that have been defined *previously*. Table 4 exemplifies this rule.

Let us assume that a block expression contains multiple function definitions for the same identifier f:

```

def f p1 = e1
    ⋮
def f pn = en

```

Then this is equivalent to

```

def f x =
  match x
  case p1 => e1
    ⋮
  case pn => en
end

```

where x is fresh for the  $p_i$  and  $e_i$ .

While only one simple definition per block and identifier is allowed, it is ok to have multiple **val**-statements for the same identifier in one block expression, for example like that:

```

val x = 1
val x = (x, x)
x

```

The above block expression evaluates to (1,1); later **val**-definitions overshadow earlier ones. But note that neither

```

val x = 1
def x = 1

```

nor

```

def x = 1
val x = 1

```

are legal.

Another difference between **val** and **def** is observed by the effects of non-determinism:

```
val x = random 2
(x, x)
```

will evaluate either to (0, 0) or to (1, 1). But

```
def x = random 2
(x, x)
```

can additionally also evaluate to (0, 1) or to (1, 0). The reason for this is that above code snippet is syntactic sugar for:

```
def x () = random 2
(x(), x())
```

*ToDo: need to explain more about order of vals and defs, especially that vals cannot reference later vals, even not indirectly via defs. This will also clean up the parts about module and object initialization.*

## 12. ANONYMOUS FUNCTIONS

So far we have seen how to define named functions in Babel-17. Sometimes we do not need a name for a certain function, for example when the code that implements this function is actually just as easy to understand as any name for the function. We already have the tools for writing such nameless, or anonymous, functions:

```
begin
  def sqr x = x * x
  sqr
end
```

is an expression denoting the function that squares its argument. There is a shorter and equivalent way of writing down the above:

```
x => x * x
```

In general, the syntax is

```
p => e
```

where  $p$  is a pattern and  $e$  an expression. The above is equivalent to

```
begin
  def f p = e
  f
end
```

where  $f$  is fresh for  $p$  and  $e$ .

There is also a syntax for anonymous functions which allows for several cases:

```
(case p1 => b1
  ⋮
  case pn => bn)
```

is equivalent to

```

begin
  def f  $p_1 = \text{begin } b_1 \text{ end}$ 
   $\vdots$ 
  def f  $p_n = \text{begin } b_n \text{ end}$ 
f
end

```

where  $f$  is fresh for the  $p_i$  and  $b_i$ .

### 13. OBJECTS

Object expressions have the following syntax:

```

object
   $s_1$ 
   $\vdots$ 
   $s_n$ 
end

```

The  $s_i$  are arbitrary statements. Those  $s_i$  that are **def** statements define the set of messages that the object responds to.

Often you do not want to create objects from scratch but by modifying other already existing objects:

```

object + parents
   $s_1$ 
   $\vdots$ 
   $s_n$ 
end

```

The expression `parents` must evaluate to either a list, a vector or a set. The members of `parents` are considered to be the parents of the newly created object, in the order induced by the collection. The idea is that the created object not only understands the messages defined by the  $s_i$ , but also the messages of the parents. Messages defined via  $s_i$  shadow messages of the parents. The messages of an earlier parent shadow the messages of a later one.

The keyword **this** has currently no meaning in Babel-17. There is one more way to denote record-like objects:

$$\{ m_1 = v_1, \dots, m_n = v_n \}$$

This is equivalent to:

```

begin
  val ( $w_1, \dots, w_n$ ) = ( $v_1, \dots, v_n$ )
  object
    def  $m_1 = w_1$ 
     $\dots$ 
    def  $m_n = w_n$ 
  end
end

```

TABLE 5. Boolean Operators

! a	a & b	a   b
<pre> <b>match</b> a   <b>case</b> <b>true</b> =&gt;     <b>false</b>   <b>case</b> <b>false</b> =&gt;     <b>true</b>   <b>case</b> _ =&gt;     <b>exception</b> TypeError <b>end</b> </pre>	<pre> <b>match</b> a   <b>case</b> <b>true</b> =&gt;     <b>match</b> b       <b>case</b> <b>true</b> =&gt;         <b>true</b>       <b>case</b> <b>false</b> =&gt;         <b>false</b>       <b>case</b> _ =&gt;         <b>exception</b> TypeError     <b>end</b>   <b>case</b> <b>false</b> =&gt;     <b>false</b>   <b>case</b> _ =&gt;     <b>exception</b> TypeError <b>end</b> </pre>	<pre> <b>match</b> a   <b>case</b> <b>false</b> =&gt;     <b>match</b> b       <b>case</b> <b>true</b> =&gt;         <b>true</b>       <b>case</b> <b>false</b> =&gt;         <b>false</b>       <b>case</b> _ =&gt;         <b>exception</b> TypeError     <b>end</b>   <b>case</b> <b>true</b> =&gt;     <b>true</b>   <b>case</b> _ =&gt;     <b>exception</b> TypeError <b>end</b> </pre>

The empty object is denoted by

**nil**

which is equivalent to

**object end**

#### 14. BOOLEAN OPERATORS

Babel-17 provides the usual boolean operators. They are just syntactic sugar for certain **match** expressions; the exact translations are given in Table 5. Babel-17 also has **if**-expressions with the following syntax:

```

if  $b_1$  then
   $e_1$ 
elseif  $b_2$  then
   $e_2$ 
   $\vdots$ 
elseif  $b_n$  then
   $e_n$ 
else
   $e_{n+1}$ 
end

```

The **elseif**-branches are optional. Also, they can be eliminated in the obvious manner via nesting, so that we only need to give the semantics for the expression

**if**  $b$  **then**  $e_1$  **else**  $e_2$  **end**

The meaning of above expression is defined to be

TABLE 6. Relational Operators

Syntax	Semantics
$a \geq b$	$b \leq a$
$a == b$	$a \leq b \ \& \ b \leq a$
$a <> b$	$!(a \leq b) \ \& \ !(b \leq a)$
$a < b$	$a \leq b \ \& \ !(a \geq b)$
$a > b$	$a \geq b \ \& \ !(a \leq b)$
$a != b$	$!(a == b)$

```

match b
  case true =>  $e_1$ 
  case false =>  $e_2$ 
  case _ => exception TypeError
end

```

## 15. ORDER

Babel-17 has a built-in partial order  $\leq$ . All other relational operators are defined in terms of this partial order (Table 6). It is possible to chain relational operators like this:

$$a \leq b \leq c > d <> e$$

The semantics of the above is

$$a \leq b \ \& \ b \leq c \ \& \ c > d \ \& \ d <> e .$$

Note that we always evaluate the operands of relational operators, even chained ones, only once. For example, the precise semantics of  $a \leq b \leq c \leq d \leq e$  is not  $a \leq b \ \& \ b \leq c \ \& \ c \leq d \ \& \ d \leq e$ , but rather

```

begin
  val t = a
  val u = b
  t <= u &
  begin
    val v = c
    u <= v &
    begin
      val w = d
      v <= w & w <= e
    end
  end
end

```

In the above, t, u, v and w are supposed to be fresh identifiers. Also note that if there are operands that are exceptions, then the result of a comparison is a dynamic exception with the same parameter as the first such operand (from left to right).

The partial order is defined by the following laws:

- $\mathbf{nil} < -\infty < z < \infty < s < l < v < cexpr < set < map < f < b$  holds for all integers  $z$ , strings  $s$ , lists  $l$ , vectors  $v$ , constructed expressions  $cexpr$ , functions  $f$ , sets  $set$ , maps  $map$  and booleans  $b$
- booleans are totally ordered and  $\mathbf{false} < \mathbf{true}$  holds
- integers are totally ordered in the obvious way
- $-\infty == -\infty$  and  $\infty == \infty$  both hold
- strings are totally ordered via lexicographic ordering item lists and vectors are partially ordered by the lexicographic ordering
- constructed expressions are partially ordered by representing them by the pair consisting of constructor and parameter
- $f < g$  holds for all functions  $f$  and  $g$
- sets are partially ordered by representing them by sorted lists that consist of the elements of the set
- maps are partially ordered by representing them by sorted lists that consist of the (key, value) pairs of the map
- let  $U$  consist of all user-defined objects (i.e., defined via **object**) that cannot receive the message **representative**; then we have  $\mathbf{nil} \leq u < -\infty$  for all  $u \in U$ ; furthermore  $U$  is partially ordered by considering its elements as sorted lists that consist of (message, value) pairs.
- let  $a$  be a user-defined object (i.e., defined via **object**) that has the message **representative**; let  $ra$  be the result of sending **representative** to  $a$ ; then  $a \leq b$  is defined by  $ra \leq b$ , and  $b \leq a$  is defined via  $b \leq ra$  for all  $b$ .

## 16. SETS AND MAPS

Sets and maps are built into Babel-17. For example, the set consisting of 3, 42 and  $\infty$  can be written as

$$\{42, \infty, 3\}$$

Sets are always sorted. The default order is Babel-17's built-in partial order, but it is also possible to create sets that work with arbitrary partial orders; an arbitrary partial order is represented as a function that takes a pair and returns a boolean. Adding or removing an element  $e$  from a set  $S$  is only well-defined when the elements of  $S$  together with  $e$  are totally ordered by the partial order of  $S$ . The same holds for testing if an element is in a set. Table 7 shows the most commonly used notation for sets.

Maps map finitely many keys to values. For example, the map that maps 1 to 2 and 4 to 0 is written as

$$\{1 \rightarrow 2, 4 \rightarrow 0\}$$

The empty map is denoted by

$$\{\rightarrow\}$$

Maps also have always a partial order associated with them, by default this is Babel-17's built-in partial order. Operations on maps are only well-defined if all keys of the map together with all other involved keys are totally ordered by the associated order. Table 8 shows the most commonly used notation for maps.

TABLE 7. Set Operations

$S + e$	set created by inserting the element $e$ into the set $S$
$S - e$	set created by removing the element $e$ from the set $S$
$S ++ T$	set created by inserting all elements of the set/list $T$ into the set $S$
$S -- T$	set created by removing all elements of the set/list $T$ from the set $S$
$S * T$	cartesian product of the set $S$ and the set $T$ , sorted by lexicographic ordering
$S ** T$	set created by removing all elements from the set $S$ which are not in the set $T$
$S e$	tests whether $e$ is an element of $S$
$S/f$	set created by applying the function $f$ to the elements of $S$
$S.\text{reorder } p$	set created by reordering $S$ according to the partial order $p$ ; in case of duplicate elements, the greater of the elements relative to the order of $S$ is kept

TABLE 8. Map Operations

$M + (k, v)$	map created from the map $M$ by associating $k$ with $v$
$M - k$	map created from the map $M$ by removing the key $k$
$M ++ N$	map created from the map $M$ by adding the key/value pairs of the map/set/list $N$
$M -- S$	map created from $M$ by removing all keys that are in the set/list/map $S$
$M * N$	map created by using the partial order of $M$ ; if $(k_M, v_M)$ is contained in $M$ and $(k_N, v_N)$ is contained in $N$ such that $k_M == k_N$ , then $(k_M, (v_M, v_N))$ is contained in the new map
$M ** T$	map created by removing all keys from the map $M$ which are not in the set/map $T$
$M k$	returns the value $v$ associated with $k$ in $M$ , or returns a dynamic exception with parameter <code>OutOfDomain</code> if no such value exists
$M \% f$	map created by applying the function $f$ to the values of $M$ ; $f$ takes a key/value pair and returns a value
$M/f$	map created by applying the function $f$ to the values of $M$ ; $f$ takes a value and returns a value
$M.\text{reorder } p$	map created by reordering $M$ according to the partial order $p$ ; in case of duplicate keys, the greater of the keys relative to the order of $M$ is kept



## 17. PATTERNS FOR SETS AND MAPS

*ToDo*

## 18. API AND RUNTIME ENVIRONMENT

Although we have already provided the specification of many operations for the built-in types of Babel-17, this document does not contain the complete API for all types yet. This API will be provided together with the reference implementation for Babel-17. Here is a quick preview of some of its features:

- Integers are arbitrary precision and support the usual operations, throwing exceptions in cases which are not well-defined.
- Division and modulo of integers are euclidean.
- Strings support the usual operations like concatenation; the basic unit is a code point, not a character. Advanced operations like the Unicode collation algorithm are not part of Babel-17 currently, but might be introduced in later versions of Babel-17.

This specification also does not contain anything about the runtime environment in which Babel-17 is supposed to run. Depending on the runtime environment, there may be additional libraries available to the programmer. The medium term goal is to have Babel-17 running on and interfacing with the following platforms:

- Java and Android
- iOS
- HTML5 / Javascript

## 19. SYNTACTIC SUGAR

One of the goals of Babel-17 is that Babel-17 code is easy to read and understand. I have found that allowing arbitrary user-specific extensions to the syntax of code are definitely not helping to achieve this goal. Nevertheless, a bare minimum of syntactic sugar is necessary to support the most basic conventions known from arithmetic; for example, one should be able to write `3 + 5` to denote the addition of the integer 3 and the integer 5.

The programmer can use some of this syntactic sugar when defining her own objects. For example, `3 + 5` is just syntactic sugar for

```
3.plus 5
```

Table 9 lists all syntactic sugar of Babel-17 that is also available to the programmer. The availability of syntactic sugar for function application means that you can let your own objects behave as if they were functions.

## 20. MEMOIZATION

Babel-17 supports *memoization*. In those places where **def**-statements can be used, **memoize**-statements can be used, also. A **memoize**-statement must always refer to **def**-statements in the same scope. Babel-17 differentiates two kinds of memoization, *strong* and *weak*. A **memoize**-statement has the following syntax:

```
memoize ref1 ... refn
```

TABLE 9. Syntactic Sugar

Sugared	Desugared
$a + b$	$a.\text{plus } b$
$a - b$	$a.\text{minus } b$
$- a$	$a.\text{uminus}$
$a * b$	$a.\text{times } b$
$a / b$	$a.\text{div } b$
$a \% b$	$a.\text{mod } b$
$a \wedge b$	$a.\text{pow } b$
$a ++ b$	$a.\text{plusplus } b$
$a -- b$	$a.\text{minusminus } b$
$a ** b$	$a.\text{timestimes } b$
$a \text{ to } b$	$a.\text{to } b$
$a \text{ downto } b$	$a.\text{downto } b$
$f \ x$	$f.\text{apply } x$

The  $\text{ref}_i$  are either of the form  $\text{id}$  to indicate strong memoization, or of the form  $(\text{id})$  to signal weak memoization. In both cases  $\text{id}$  refers to the **def**-statement being memoized. As an example, here is the definition of the weakly memoized fibonacci-sequence:

```

memoize (fib)
def fib 0 = 0
def fib 1 = 1
def fib n = fib (n-1) + fib (n-2)

```

The difference between weak and strong memoization is that strong memoization guarantees that the memoized function will not be executed twice for the same argument; weak memoization does not make this guarantee, so that garbage collection can reclaim all or part of the memoization cache when it needs to. Note that all arguments to the same memoized function must be totally ordered by the built-in partial order.

## 21. LINEAR SCOPE

We extend now the syntax of **val**-statements such that it is actually legal to leave out the keyword **val** if all variables defined by it have already been defined previously in its surrounding *linear scope*. The following two expressions are equivalent:

<pre> <b>begin</b>   <b>val</b> x = a   <b>val</b> y = b   <b>val</b> x = c   d <b>end</b> </pre>	<pre> <b>begin</b>   <b>val</b> x = a   <b>val</b> y = b   x = c   d <b>end</b> </pre>
---	--

Furthermore, **begin** ... **end** can not only be used as an expression, but also as a statement; the following expressions exemplify nested **begin** ... **end** statements:

<pre> begin   val x = 1   val y = 2   begin     val x = 3     val y = 4 * x   end   (x, y) end </pre>	<pre> begin   val x = 1   val y = 2   begin     val x = 3     y = 4 * x   end   (x, y) end </pre>	<pre> begin   val x = 1   val y = 2   begin     val x = 3     val y = 0     y = 4 * x   end   (x, y) end </pre>
evaluates to (1, 2)	evaluates to (1, 12)	evaluates to (1, 2)

The above examples show that the effect of dropping the **val** in a **val** statement is that the binding of an identifier becomes visible at that level within the linear scope where it has last been introduced via a **val** statement that includes the **val** keyword or via a pattern match in a **match** or **def**.

The linear scope of a statement is defined to be the block expression  $b$  that directly contains the statement, together with the linear scope of the expression or statement that contains  $b$ . The linear scope of an expression  $e$  is only defined when  $e$  is a *control expression* that appears in another **val** statement of the form

**val**  $pat = e$

or

$pat = e$

Then the linear scope of  $e$  is defined as the linear scope of that **val** statement. These are all *control expressions* of Babel-17:

- **begin ... end**
- **if ... end**
- **match ... end**
- **for ... end**
- **while ... end**

The last two control expressions are loops and explained in the next section.

Here are three example code snippets that illustrate linear scope:

<pre> begin   val x = 1   val y =     begin       x = 2       x+x     end   (x, y) end </pre>	<pre> begin   val x = 1   val y = 3 *     begin       x = 2       x+x     end   (x, y) end </pre>	<pre> begin   val x = 1   val y = 3 *     begin       val x = 2       x+x     end   (x, y) end </pre>
evaluates to (2, 4)	illegal	evaluates to (1, 12)

Just as the **begin ... end** expression may be used as a statement, you can also use all other control expressions as statements, for example:

```
begin
  val x = random 2
  if x == 0 then
    x = 100
  else
    x = 200
  end
  x + x
end
```

This expression will evaluate either to 200 or to 400.

For **if**-statements the **else**-branch is optional, and **match**-statements do not throw a **NoMatch** exception if none of the patterns matches.

## 22. YIELD

Control expressions always return a value. With the help of the **yield** statement and, optionally, *collector annotations*, the programmer has flexible control about how this value is computed. The syntax of a **yield** statement is simply

```
yield e
```

where  $e$  is an arbitrary expression. The semantics is that the value of the control expression is the collection of all the values that have been issued via **yield**. For example, the expression

```
begin
  yield 1
  val x = 2
  yield x*x
  yield 3
  10
end
```

evaluates to (1, 4, 3, 10). Note how the normal return value of **begin ... end** just counts as an additional **yield**.

By default, all yielded values are collected in a list which collapses in case of only a single element. The programmer might want to deviate from this default and collect the yielded values differently, for example to get rid of the collapsing behavior. *Collector annotations* allow her to do just that. A collector annotation has the following syntax:

```
with c : e
```

where  $c$  is a *collector* and  $e$  is a control expression. A collector is any object that

- responds to the message `collector_close`,
- and responds to the message `collector_add x` with another collector.

Lists, sets and maps are built-in collectors which the programmer can use out-of-the-box; apart from that she can implement her own collectors, of course.

Here is an example where we use a set as a collector:

```

with {4} : begin
  yield 1
  yield 2
  yield 1
  10
end

```

Above expression evaluates to {1, 2, 4, 10}.

## 23. LOOPS

Unlike the other control expressions, *loops* become useful only in combination with *linear scope* and **yield**.

This is the syntax for the **while**-loop:

```

while c do
  b
end

```

Here *c* must evaluate to a boolean and *b* is a control block. For example, here is how you could code the euclidean algorithm for calculating the greatest common divisor using **while**:

```

def gcd (a,b) = begin
  while b != 0 do
    (a, b) = (b, a mod b)
  end
  a
end

```

There is also the **for**-loop. It has the following syntax:

```

for p in C do
  b
end

```

In the above *p* is a pattern, *C* is a *collection* *C*, and *b* is a control block. The idea is that above expression iterates through those elements of the *collection* *C* which match *p*; for each successfully matched element, *b* is executed.

An object *C* is a collection if it handles the message **iterate**

- by returning () if it represents the empty collection,
- by returning (*e*, *C'*) such that *C'* is also a collection otherwise.

Here is an example of a simple **for**-loop expression:

```

begin
  val s = [10, (5, 8), 7, (3,5)]
  with {->} : for (a,b) in s do
    yield (b,a)
  end
end

```

evaluates to  $\{8 \rightarrow 5, 5 \rightarrow 3\}$ .

Using **for**-loops it is possible to formulate all of those *fold*-related functionals known from functional programming in a way which is easier to parse for most people.